globals 
[
  numAbs  ;; the number of antibodies secreted by plasma cells at each time step
  numCyto ;; the number of cytokines produced by effector T helper cells at each time step
  entities ;; the number of all non-viral entities in the simulation
  life_B  ;; half-life of a B cell
  life_Plasma ;; half-life of a plasma cell 
  life_MemB ;; the half-life of a memory B cell
  life_Th  ;; the half-life of a naive T helper cell
  life_MemTh  ;; the half-life of a memory T helper cell
  life_EffTh  ;; the half-life of an effector T helper cell
  life_Tc ;; the half-life of a naive T cytotoxic (Tc) cell
  life_MemTc  ;; the half-life of a memory Tc cell
  life_EffTc  ;; the half-life of an effector Tc cell
  life_Ab  ;; the half-life of an antibody
  life_Cyto  ;; the half-life of a cytokine
  life_Apc  ;; the half-life of an antigen-presenting cell (apc)
  life_Virion ;; the half-life of a virion - an extracellular viral particle
  cytokines
  plasmacells
  epcells
  infected ;; keeps track of the number of infected cells at any given time during the simulation
  neutralized ;; keeps track of the total number of virions neutralized by antibodies
  killed ;;  keeps track of the total number of infected cells killed by effector Tc cells
  pdissociation ;; the probability that an APC or, in this case, a follicular denritic cell bound to hiv will dissociate at a given time step
  p_nonspec ;; probability of successful binding in a nonspecific interaction, e.g., APC/Ag
  bystander_thr
  max_sti ;; the number of stimulations needed to render a B cell anergic
  life_anergy ;; the half-life of the anergic state
  mutB ;; the B cell hypermutation rate
  mutV ;; the viral mutation rate
  maxspec ;; the upper bound of the numerical values for the receptors, epitoptes, paratopes, etc, used in the simulation.
]

breed [ bCell ]
breed [ memB ]
breed [ plasma ]
breed [ thCell ]
breed [ memTh ]
breed [ effTh ]
breed [ tcCell ]
breed [ memTc ]
breed [ effTc ]
breed [ ab ]
breed [ apc ]
breed [ hiv ]
breed [ cyto ]
breed [ epCell ]
breed [ ara ]


turtles-own 
[
  stimulated? 
  state 
  infected? 
  halfL ;; half-life
  numDivs 
  sinceInf ;; the number of time steps since the given turtle has been infected. The value of this variable, 
           ;; together with the viral replication rate, is used to calculate the number of viral particles present in a 
           ;; productively infected cell at any given time, post infection
  partner  ;; the virion with which this turtle is complexed, if any
  HLA1/pep
  HLA2/pep
  receptor ;; the numeric value of this cell's receptor
  inf_lethal_load ;; the viral load of the virion with which this cell is infected, if applicable
  inf_lethality ;; the infectivity/lethality of the virion with which this cell is infected, if applicable
  inf_rep_rate  ;; the replication rate of the virion with which this cell is infected, if applicable
  vLoad ;; viral load
]

patches-own
[
  cytos ;; the number of cytokines on a given patch - used to simulating chemotaxis
  inf ;; the number of infected cells on a given patch - used to simulate chemotaxis
]

hiv-own 
[
  inactivated? ;; determines if the virion has been inactivated by an anti-retroviral agent - inactivated virions are incapable of replication of replication, following infection of a target cell
  lethality ;; the probability that a virion will infect a cell on contact
  lethal_load ;; the threshold of viral particles that elicits the lysis of an infected cell 
  rep_rate ;; the replication rate of this virion (in an infected cell)
  peptide ;; the numerical value of the viral peptides that is presented on HLA1/peptide and HLA2/peptide complexes
  epitope ;; the numerical value of the epitope of a virion
]
ab-own
[
  paratope
]
ara-own 
[ 
  effect ;; the effectiveness of an anti-retroviral agent
]
apc-own
[
  vRNA
]

to setup
  ;; (for this model to work with NetLogo's new plotting features,
  ;; __clear-all-and-reset-ticks should be replaced with clear-all at
  ;; the beginning of your setup procedure and reset-ticks at the end
  ;; of the procedure.)
  clear-all
  reset-ticks
;  random-seed 250
  initialize-globals
  initialize-turtle-shapes
	create-cells
	initialize-patch-vars
	update-monitors
end

to initialize-globals
	set life_B 15
	set life_Plasma 5
	set life_MemB 10
	set life_Th 15
	set life_MemTh 10
	set life_EffTh 5
	set life_Tc 15
	set life_MemTc 10
	set life_EffTc 5
	set life_Ab 2
	set life_Cyto 2
	set life_Apc 15
	set life_Virion 1000
	set numAbs 2
	set numCyto 2
	set neutralized 0
	set killed 0
	set pdissociation 0.001
	set p_nonspec .5
	set bystander_thr 5
	set life_anergy 10
	set mutV 0
	set mutB 0
	set maxspec 3
end

to initialize-patch-vars
  ask patches
  [
    set pcolor 16
    set inf (count turtles-here with [ infected? ])
  ]
end  

to update ;  startup
	ask turtles with [ breed != hiv and breed != epCell ] [ without-interruption [ do-interact] ]
	ask turtles with [ infected? ] [ do-viral-production ]
	ask turtles [ do-apoptosis ]
	ask ab [ do-ab-neutralization ]
	ask plasma [ do-ab-secretion ]
	ask effTh [ do-cyto-production ]
	ask turtles with [ stimulated? and not infected? and (breed != apc) ] [ do-cell-division ]
	ask effTc [ do-ctl ]
	ask hiv [ do-infection ]
	ask apc with [ state = "stimulated" ] [ do-dissociation ]
	do-resupply
	ask turtles with [ breed = hiv or breed = cyto ] [ do-diffuse ]
	update-patch-vars
	ask turtles with [ breed != hiv and breed != cyto and breed != epCell ] [ do-diffuse ]
	ask ara [ do-ART ]
	ask apc [ do-dissociation ]
	update-monitors
  tick
end


;; this procedure can be used to simulate antibody therapy; simply provide the specificity and number of
;; antibodies you wish to introduce into the simulation. Note that you can use the real-time information about the dominant epitope 
;; to guide you in your choice of specificity and quantity of antobodies to introduce.
to simulate-ABT [specificity number]
  create-ab number
  [
    initialize-turtle-vars
    set paratope specificity    
  ]
  show (word number " new antibodies added at time step " ticks)
end

;; this procedure can be used to simulate anti-retriviral therapy; simply provide the specificity, effectiveness, and number of
;; anti-retroviral agents you wish to introduce into the simulation
to simulate-ART [efficacy number]
  create-ara number
  [
    initialize-turtle-vars
    set effect efficacy
  ]
  show (word number " new anti-retroviral agents added at time step " ticks)
end

;; this procedure can be used to introduce into the simulation new virions of a particular specificity
to infect [ specificity leth bsize repRate number ]
  create-hiv number
  [
    initialize-turtle-vars
    set receptor specificity
    set epitope specificity
    set lethality leth
    set lethal_load bsize
    set rep_rate repRate    
  ]
  show (word number " new virions added at time step " ticks)
end

to update-patch-vars
  ask patches 
  [ 
    set inf (count turtles-here with [ infected? ]) 
    set cytos (count cyto-here) 
  ]
end

to do-ART
	let _other 0
  
	set _other one-of hiv in-radius 0.2
	if((_other != nobody) and (prob effect)) [ ask _other [ die ] ]
end	
		
to update-monitors
  set entities (count turtles with [ breed != hiv ])
  set cytokines (count cyto)
  set plasmacells (count plasma)
  set infected get-infected
  set epcells (count epCell)
end

to-report get-infected
  let diff 0
  
  set diff 0
  if(not any? turtles with [ infected? ]) [ report "0 / 0" ]
  foreach [vLoad] of (turtles with [ infected? ])
  [
    set diff (diff + ?)
  ]
  report (word (count turtles with [ infected? ]) " / " int(diff))
end

to-report get-dom-epitope
  if(count hiv = 0) [ report "------ NONE ------" ]
  
  let maximum 0
  let value 0
  let mspec 0
  let aspec 0
  
  set aspec 0
  set maximum 0
  set value 0
  while [ aspec < maxspec ]
  [
    set value count (hiv with [ epitope = aspec ])
    if(value > maximum) 
    [ 
      set maximum value 
      set mspec aspec
    ]
    set aspec (aspec + 1)
  ]
  
  let antibodies (count ab)
  report (word mspec "           |   " maximum "       |   " int((maximum / (max (list 1 count hiv))) * 100) "%" "     |   " count (ab with [ exp(0 - abs(receptor - mspec)) >= .7 ]))
end

to initialize-turtle-shapes
  set-default-shape turtles "circle"
end
 
to initialize-turtle-vars
	set infected? false
	set stimulated? false
	set state "resting"
	set numDivs 0
	set receptor random maxspec
	set sinceInf 0
	set partner nobody
	set halfL 1000
	if(breed = hiv) 
	[ 
	  set inactivated? false 
	  set color black 
	  set peptide random maxspec 
	  set epitope peptide 
	  set lethality infectivity 
	  set lethal_load 24 
	  set rep_rate 0.05 
	]
	if(breed = bCell) 
	[ 
	  set halfL life_B 
	  set color white 
	]
	if(breed = plasma) 
	[ 
	  set halfL life_Plasma 
	  set color green 
	]
	if(breed = memB) 
	[ 
	  set halfL life_MemB 
	  set color white
	]
	if(breed = thCell) 
	[ 
	  set halfL life_Th 
	  set color blue 
	]
	if(breed = memTh) 
	[ 
	  set halfL life_MemTh 
	  set color blue 
	]
	if(breed = effTh) 
	[ 
	  set halfL life_EffTh 
	  set color green 
	]
	if(breed = epCell) 
	[ 
	  set halfL 1000 
	  set color green 
	]
	if(breed = tcCell) 
	[ 
	  set halfL life_Tc 
	  set color blue 
	]
	if(breed = memTc) 
	[ 
	  set halfL life_MemTc 
	  set color blue 
	]
	if(breed = effTc) 
	[ 
	  set halfL life_EffTc 
	  set color green 
	]
	if(breed = ab) 
	[ 
	  set halfL life_Ab 
	  set color yellow
	]
	if(breed = cyto) 
	[ 
	  set halfL life_Cyto 
	  set color brown
	]
	if(breed = apc) 
	[ 
	  set halfL life_Apc 
	  set vRNA []
	  set color grey
	]
end

to create-cells
	create-bCell numB
	[
		initialize-turtle-vars
		disperse
	]
	create-epCell numEp
	[
		initialize-turtle-vars
		disperse
	]
	create-hiv numHiv
	[
		initialize-turtle-vars
		disperse
	]
	create-thCell numTh
	[
		initialize-turtle-vars
		disperse
	]
	create-apc numApc
	[
		initialize-turtle-vars
		disperse
	]
	create-tcCell numTc
	[
		initialize-turtle-vars
		disperse
	]
end

;; the interactions of entities, as described in the supplemental materials
to do-interact
	let this 0
  let _other 0
  
	set this self
	if ((breed = bCell or breed = memB) and (state = "resting"))
	[
		set _other one-of hiv in-radius .2 with [ true ]
		if((_other != nobody) and (success receptor ([epitope] of _other)))
		[
			set state "activated" ;; the B cell is still in resting state but is presenting an HLA2/peptide complex. 
			                    ;; Subsequent binding with a Th cell will result in its activation
			set HLA2/pep ([peptide] of _other)
			ask _other [ die ]
		]
	]
	if((breed = bCell or breed = memB) and not stimulated?)
	[
		if((count cyto-here) >= bystander_thr) ;; bystander activation of B cells and memory B cells
		[
		  set state "activated"
			set stimulated? true
		]
	]
	if((breed = bCell or breed = memB) and not stimulated? and (state = "activated"))
	[
		set _other one-of turtles in-radius .2 with [ ((breed = thCell) or (breed = memTh)) ]
		if(_other != nobody  and (success HLA2/pep ([receptor] of _other)))
		[
			set stimulated? true
;			set [stimulated?] of _other true ;; B cells are also APCs and can thus stimulate Th cells
ask _other [set stimulated? true]  ; rpmcruz
		]
	]
	if(((breed = thCell) or (breed = memTh)) and not stimulated?)
	[
		set _other one-of turtles in-radius .2 with [ ((breed = apc)) and (state = "activated") ]
		if((_other != nobody)  and (success receptor ([HLA2/pep] of _other)))
		[
			set state "activated"
			set stimulated? true
;			set [stimulated?] of _other true ;; this has no effect on apcs. However, stimulated B cells are able to divide
ask _other [set stimulated? true]  ; rpmcruz
		]
	]
	if(((breed = tcCell) or (breed = memTc)) and not stimulated?)
	[
		set _other one-of apc in-radius .2 with [ stimulated? ]
		if((_other != nobody) and (success receptor ([HLA1/pep] of _other)))
		[
			set state "activated"
			set stimulated? true
		]
	]
	if(((breed = tcCell) or (breed = memTc)) and not stimulated?)
	[
		if((count cyto-here) >= bystander_thr)
		[
		  set state "activated"
			set stimulated? true
		]
	]
	if(breed = apc) ;; apcs (particularly folicular dendritic cells) may serve as viral reservoirs
	[
		set _other one-of hiv in-radius .2 with [ true ]
		if((_other != nobody) and prob p_nonspec)
		[
			set state "activated"
			set HLA1/pep ([peptide] of _other) ;; an activated apc presents viral peptides on both HLA1 and HLA2
			set HLA2/pep ([peptide] of _other)
			set vRNA lput ([peptide] of _other) vRNA
			ask _other [ die ]			
		]
	]
end

;; the half-life of each entity is used to determine the probability that it will die during a given time step. Because the gradual 
;; depletion of Th cells, a hallmark of hiv pathogenesis, is thought to arise not only from direct hiv-mediated killing of infected 
;; Th cells but also from indirect killing (or bystander effect) of Th cells. We assume 
;; that this indirect killing of Th cells is a function of the number of both activated and infected cells in the vicinity of a 
;; Th cell and on the value of the AICD parameter.
to do-apoptosis
	let diff 0
  let tau 0
  
	if((breed = thCell) or (breed = memTh) or (breed = effTh)) [ set halfL max (list 0 (halfL - (aicd * ((count turtles-here with [ infected? ]) + (count hiv-here))))) ] 
	set tau halfL
	set diff ((ln 2) * tau)
	if(prob exp (0 - diff)) [ die ]
end
	
;; the dissociation of an apc from a bound virion. This is thought to contribute to the pepertuation of viremia in chronically 
;; infected patients.
to do-dissociation
  let this 0
  
  set this self
  if(prob pdissociation and not empty? vRNA)
  [
    foreach vRNA
    [
      hatch 1
      [
        set breed hiv
        initialize-turtle-vars
        set epitope ?
        set peptide ?
      ]
    ]
    set vRNA [ ]
    set state "resting"
    set stimulated? false
  ]
end

;; the neutralization of a virion by an antibody. Note that to keep the number of turtles relatively low and, consequently, improve the 
;; speed of the simulation, all neutralized virions and the antibodies involved are immediately removed from the simulation.
to do-ab-neutralization
	let _other 0
  let this 0
  
	set this self
	set _other one-of hiv in-radius .2 with [ success epitope ([paratope] of this) ]
	if((_other != nobody) and (prob neutRate))
	[ 
		set neutralized (neutralized + 1)
		ask _other [ die ]
		die
	]
end

;; viral infection of target cells. The probability of infection is determined by the infectivity parameter or the lethality variable
to do-infection
	let this 0
  let _other 0
  
	set this self
	set _other one-of turtles in-radius .2 with [ (breed = thCell) or (breed = memth) or (breed = effTh) ] ;; hiv preferentially infect th cells
	if(_other = nobody) [ set _other one-of turtles in-radius .2 with [ (breed != hiv) and (breed != cyto) and (breed != ab) and (breed != ara)] ]
	if(_other != nobody)
	[
	  if(prob lethality)
	  [
;	      set [infected?] of _other true
;		    set [HLA1/pep] of _other peptide
;		    set [inf_rep_rate] of _other rep_rate ;; the replication rate for inactivated virions, in vivo, is set to 0
;		    set [inf_lethality] of _other lethality
;		    set [inf_lethal_load] of _other lethal_load
; rpmcruz
let p peptide
let rr rep_rate
let l lethality
let ll lethal_load
ask _other [
  set infected? true
  set HLA1/pep p
  set inf_rep_rate rr
  set inf_lethality l
  set inf_lethal_load ll
]
		    die
		]
	]
end

;; plasma cells secrete numAbs antibodies during each time step
to do-ab-secretion
	let this 0
  
	set this self	
	hatch numAbs
	[
		set breed ab
		initialize-turtle-vars
		set paratope ([receptor] of this)
	]
end

;; effector Th cells secrete numCyto cytokines during each time step
to do-cyto-production
	let this 0
  
	set this self
	hatch numCyto
	[
		set breed cyto
		initialize-turtle-vars
	]
end
	
;; the killing of infected cells by Tc cells. The probability that a Tc cell will kill an infected cell on initial contact is given 
;; by the ctlRate parameter
to do-ctl
	let _other 0
  
	set _other one-of turtles in-radius .2 with [ infected? ]
	if((_other != nobody) and (success receptor ([HLA1/pep] of _other)) and (prob ctlRate)) [ ask _other [ die ] set killed (killed + 1)]
end
	
;; activated B, memory B, Th, memory Th, Tc, and memory Tc cells divide over two time steps to produce 4 cells - 2 effectors cells and
;; 2 memory cells
to do-cell-division
	let this 0
  
	set this self
	ifelse(numDivs >= 2)
	[
		if((breed = bCell) or (breed = memB))
		[
			hatch 2
			[
				set breed memB
				initialize-turtle-vars
				ifelse(prob mutB) [ set receptor random maxspec ] [ set receptor ([receptor] of self) ]
			]
			hatch 2
			[
				set breed plasma
				initialize-turtle-vars
				set receptor ([receptor] of this)
			]
			die
		]
		if((breed = thCell) or (breed = memTh))
		[
			hatch 2
			[
				set breed memTh
				initialize-turtle-vars
				set receptor ([receptor] of this)
			]
			hatch 2
			[
				set breed effTh
				initialize-turtle-vars
				set receptor ([receptor] of this)
			]
			die
		]
		if((breed = tcCell) or (breed = memTc))
		[
			hatch 2
			[
				set breed memTc
				initialize-turtle-vars
				set receptor ([receptor] of this)
			]
			hatch 2
			[
				set breed effTc
				initialize-turtle-vars
				set receptor ([receptor] of this)
			]
			die
		]
	]
	[
		set numDivs (numDivs + 1)
	]
end

;; productively infected cells (as opposed to infected but dormant/anergic cells)	continue to produce new virions until
;; either a) they are killed during CTL or (b) their viral load exceeds the lethal load of the infecting virus. In the latter
;; case, the infected cells undergo lysis, releasing new virions into the simulation.
to do-viral-production
	let this 0
  
	set this self
	set vLoad floor(exp(inf_rep_rate * sinceInf))
	ifelse( vLoad > inf_lethal_load)
	[
		hatch vLoad
		[
			set breed hiv
			initialize-turtle-vars
			ifelse(prob mutV) 
			[ 
			  set peptide random maxspec 
			  set epitope random maxspec 
		  ] 
			[ 
			  set peptide HLA1/pep 
			  set epitope HLA1/pep 
			]
			set lethal_load inf_lethal_load 
			set rep_rate inf_rep_rate 
			set lethality inf_lethality 
		]
		die
	]
	[
		set sinceInf (sinceInf + 1)
	]
end

;; to maintain a relatively constant supply of naive lymphocytes, these cells are replenished by the bone marrow and thymus at the same rate
;; as they undergo apoptotic death	
to do-resupply
	let diff 0
  let tau 0
  
	set diff 0
	while [ diff < numTh ]
	[
	  if(prob exp (- ((ln 2) * life_Th)))
	  [
		  create-thCell 1
		  [
			  initialize-turtle-vars
		  ]
	  ]
	  set diff (diff + 1)
	]
	set diff 0
	while [ diff < numTc ]
	[
	  if(prob exp (- ((ln 2) * life_Tc)))
	  [
		  create-tcCell 1
		  [
			  initialize-turtle-vars
		  ]
	  ]
	  set diff (diff + 1)
	]
	set diff 0
	while [ diff < numB ]
	[
	  if(prob exp (- ((ln 2) * life_B)))
	  [
		  create-bCell 1
		  [
			  initialize-turtle-vars
		  ]
	  ]
	  set diff (diff + 1)
	]
	set diff 0
	while [ diff < numApc ]
	[
	  if(prob exp (- ((ln 2) * life_Apc)))
	  [
		  create-apc 1
		  [
			  initialize-turtle-vars
		  ]
	  ]
	  set diff (diff + 1)
	]
end

;; an alternate implementation of the resupply process
to do-resupply2
  let tauB 0
  let tauTc 0
  let tauTh 0
  
  set tauB ((ln 2) * numB / life_B)
  set tauTh ((ln 2) * numTh / life_Th)
  set tauTc ((ln 2) * numTc / life_Tc)
	create-bCell ceiling(tauB)
	[
	  initialize-turtle-vars
	  disperse
	]  
	create-thCell ceiling(tauTh)
	[
	  initialize-turtle-vars
	  disperse
	]  
	create-tcCell ceiling(tauTc)
	[
	  initialize-turtle-vars
	  disperse
	]
end

to disperse
	setxy random world-width random world-height
end

;; the dissufion of entities is assumed to be driven by a chemotactic gradient. For example, naive Tc cells diffuse in the 
;; direction of greatest cytokine concentration, while effector Tc cells seek out sites with high numbers of infected cells.
to do-diffuse
  repeat 2
  [
	  ifelse(breed = tcCell or breed = memTc or breed = bCell or breed = memB) 
	  [ 
;	    set heading uphill cytos
  face max-one-of neighbors [cytos]  ; rpmcruz
	  ]
	  [
	    ifelse(breed = effTc) 
	    [ 
;	      set heading uphill inf 
  face max-one-of neighbors [inf]  ; rpmcruz
	    ]
	    [
	      set heading random 360
	    ]
	  ]
	  fd 1
	  if (partner != nobody) 
	  [ 
;	    set [heading] of partner heading 
let h heading
ask partner [set heading h]

	    ask partner [ fd 1 ] 
	  ]
	]
end
	
;; this procedure determines the probability of successful interaction (e.g., bond formation) between two entities
to-report success [ receptor1 receptor2 ]
	let var 0
  
	ifelse((not is-number? receptor1) or (not is-number? receptor2)) 
	[ 
	  report false
	]
	[
	  set var abs(receptor1 - receptor2)
	  ifelse(prob exp (0 - var)) [ report true ] [ report false ]
	]
end

;; this procedure determines the outcome of events (i.e. true or false) based on the probabilities of the events
to-report prob [ probability ]
	let var 0
  
	if(probability = 0) [ report false ]
	set var (1 / probability)
	ifelse(random-float var < 1.0) [ report true ] [ report false ]
end
@#$#@#$#@
GRAPHICS-WINDOW
379
13
717
372
20
20
8.0
1
10
1
1
1
0
1
1
1
-20
20
-20
20
0
0
1
ticks
30.0

PLOT
440
387
733
598
HIV/Th/Tc Dynamics
Time Step
NIL
0.0
100.0
0.0
100.0
true
true
"" ""
PENS
"th" 1.0 0 -16777216 true "" "plot (count thCell + count memTh + count effTh)"
"hiv" 1.0 0 -65536 true "" "plot count hiv"
"tc" 1.0 0 -16776961 true "" "plot (count tcCell + count memTc + count effTc)"

BUTTON
11
13
80
75
Setup
setup
NIL
1
T
OBSERVER
NIL
NIL
NIL
NIL
1

BUTTON
93
15
166
76
Run
update
T
1
T
OBSERVER
NIL
NIL
NIL
NIL
1

SLIDER
10
112
170
145
numB
numB
0
100
50
1
1
NIL
HORIZONTAL

SLIDER
183
112
337
145
numTh
numTh
0
100
50
1
1
NIL
HORIZONTAL

SLIDER
11
153
170
186
numTc
numTc
0
100
50
1
1
NIL
HORIZONTAL

SLIDER
183
153
337
186
numApc
numApc
0
100
55
1
1
NIL
HORIZONTAL

SLIDER
183
195
337
228
numHiv
numHiv
0
300
50
1
1
NIL
HORIZONTAL

SLIDER
13
195
170
228
numEp
numEp
0
1000
500
1
1
NIL
HORIZONTAL

SLIDER
13
235
168
268
ctlRate
ctlRate
0
1
0.85
0.0010
1
NIL
HORIZONTAL

SLIDER
14
279
169
312
infectivity
infectivity
0
1
0.007
0.0010
1
NIL
HORIZONTAL

SLIDER
183
278
336
311
AICD
AICD
0
5
0.11
0.01
1
NIL
HORIZONTAL

MONITOR
18
400
280
445
Epitope   |    No.     |   Prop.    |   Abs
get-dom-epitope
3
1
11

SLIDER
183
236
336
269
neutRate
neutRate
0
1
0.109
0.0010
1
NIL
HORIZONTAL

MONITOR
15
340
121
385
Non-Viral Entities
count turtles with [ breed != hiv ]
3
1
11

MONITOR
127
341
276
386
Infected Cells/ Viral Load
get-infected
3
1
11

MONITOR
114
460
207
505
Neutralized HIV
neutralized
3
1
11

MONITOR
18
460
101
505
Killed
killed
3
1
11

PLOT
766
16
990
187
Antibodies
Time Step
NIL
0.0
100.0
0.0
100.0
true
false
"" ""
PENS
"Antibodies" 1.0 0 -13345367 true "" "plot (count ab)"

PLOT
767
401
992
592
Bcells
Time Step
NIL
0.0
100.0
0.0
100.0
true
false
"" ""
PENS
"Bcells" 1.0 0 -2674135 true "" "plot (count bCell + count memB + count plasma)"

PLOT
768
205
992
383
cytokines
Time Step
NIL
0.0
100.0
0.0
100.0
true
false
"" ""
PENS
"Cytokines" 1.0 0 -10899396 true "" "plot (count cyto)"

@#$#@#$#@
## WHAT IS IT?

HIVSIM is an agent-based simulation of HIV immunodynamics that is currently being developped in NetLogo. It allows users to investigate dependencies between various components of the cellular and humoral immune responses to HIV. Users can interactively manipulate simulation parameters (e.g., the number of Th, Tc, and B cells, and the infectivity of viral particles) and, in real-time, observe graphical plots of the results. Additionally, users can simulate antibody and anti-retroviral therapies at various stages of infection (e.g., the user can introduce into the simulation antibodies with affinity for the dominant HIV epitope). HIVSIM is still a work in progress. As time permits, the underlying model will be further calibrated against experimental data to make it robust and applicable to the qualitative evaluation of hypotheses on HIV immunodynamics.

## HOW IT WORKS

The simulation  
HIVSIM simulates the interactions of five major cell types - B, T helper, T cytotoxic, epithelial, and antigen-presenting cells - and four non-cellular entities - antibodies, cytokines, HIV, and anti-retroviral agents. The interactions occur in a cellular matrix consisting of 1681 (41 * 41) sites. During each time step, all entities at a given site are allowed to interact. After the interactions, other processes such as cell division, apoptosis or decay, antibody and cytokine production, necrosis, and HIV infection of target cells are allowed to proceed. The completion of these processes marks the end of a time step, after which all entities, except epithelial cells, are allowed to diffuse into neighboring sites. HIV particles, antibodies and anti-retroviral agents can be introduced from the outside at any time. The simulation ends when one of the following occurs (1) the number of healthy epithelial cells drops to below 50% of its original value, (2) the viral load exceeds the combined total of all non-viral entities, or (3) the viral load falls to 0.

Described below are some of the entities, interactions, and biological processes implemented in HIVSIM. Except where otherwise noted, information regarding the various interactions and processes came from [1]. Also, some of the parameters used in the simulation were adapted from [2]. For detailed information about the simulation, please see comments within the code.

Let us first of all discuss the prerequisites for a "successul interaction" between two entities. Note that except for nonspecifc interactions such as those involving APCs and cell-free HIV, the outcome (e.g., successful formation of a bond) of most interactions depends on the receptors (or epitopes and paratopes) of the entities involved. Each cellular entity possesses a receptor, represented by a randomly chosen integer belonging to the closed-open interval [0, maxspec), where the upper bound maxspec is determined by the user. Cellular entities also possess HLA1 (human leukocyte antigen type 1) with similar numeric values as their receptors. In additon to HLA1, APCs and B cells also have HLA2. Among the non-cellular entities, viruses have epitopes while antibodies have paratopes. When two entities A and B, say, with receptors x and y, respectively, interact, the probability that their interaction will be successful is given by exp(|x-y|).  The closer the numeric values of the interacting entities' receptors, the higher their affinities for each other and, hence, the higher the probability of a successful interaction. 

Entities and interactions  
B cells: B cells are components of the humoral immune response to HIV. When a naive B cell encounters an HIV particle, it attempts to bind to it. Successful binding of a B cell to HIV results in the former ingesting the latter. After ingestion, the B cell digests the HIV into small peptides, 11 to 20 amino acids long, and inserts these into molecules called human leukocyte antigen type II (HLA2), forming HLA2-peptide complexes. It then transports these HLA2-peptide complexes to its surface. Th cells, which will be described later, can recognize and bind to the HLA2-peptide complexes. Such binding activates B and Th cells and induces these cells to divide (see "Cell Division" below). The division of B cells yields memory and effector (or plasma) cells. The long-lived memory B cells confer long-term protection against previously encountered HIV particles. When a memory B cell becomes activated (e.g. via a successful interaction with an APC-activated Th cell or by-stander, cytokine-induced activation), it undergoes cell division, increasing the population of plasma and memory B cells with affinity for the HIV. The short-lived and terminally differentiated plasma cells continuously produce antibodies which can recognize and bind to cell-free HIV. Such binding can neutralize the HIV, rendering it incapable of infecting target cells in the body.

T helper (Th) cells: Also known as CD4+ T cells, these cells are the orchestrators of the adaptive immune response to HIV. When a naive Th cell binds to HIV peptides presented on HLA2 surface molecules of antigen presenting cells (APCs) they become activated. Activated Th cells divide to produce memory and effector Th cells (see "Cell Division" below). The short-lived effector Th cells continuously secrete cytokines which participate in the activation of other immune system components including B cells and Tc cells (see below). Upon activation, the long-lived memory Th cells divide, as described below, producing more memory and effector Th cells. Th cells also facilitate the lysis of infected cells by natural killer (NK) cells in a process called antibody dependent cell-mediated cytotoxicity. The gradual depletion of Th cells, a hallmark of HIV pathogenesis, is thought to arise not only from direct hiv-mediated killing of infected Th cells but also from indirect killing of Th cells or activation-induced cell death (AICD) [3]. It is assumed here that this indirect killing of Th cells is a function of the number of infected cells and virions in the vicinity of a Th cell. The parameter AICD is used to simulate this increased susceptibility of Th cells to HIV.

T cytotoxic (Tc) cells: Tc cells are very important in the control of HIV infection. They are activated either by direct contact with APC-activated Th cells or by cytokines. Once activated, they divide and produce memory and effector Tc cells. Effector Tc cells recognize HIV peptides, 9 to 11 nucleotides in length, bound to molecules called human leukocyte antigens type 1 (HLA1) found on most human cell types. Effector Tc cells bind to and kill infected cells displaying such HLA1-peptide complexes on their surfaces, at a rate that is determined by the ctl-rate parameter.Memory Tc cells, on the other hand, afford long-term protection against infection. Upon activation (see naive Tc cell activation) the memory cells divide, as described below.

Antigen-presenting cells (APCs): These include macrophages, dendrites and lymphocytes. All APCs possess HLA2 molecules. Note that if Th cell activation was exclusively dependent on B cell activation, the primary response to HIV would be very slow due to the initial low numbers of B and T cells with affinity for the HIV pathogens. Fortunately, that is not the case; APCs other than B cells can bind non-specifically to HIV. After binding, the APCs ingest, digest and present HLA2-peptide complexes on their surfaces. The binding of Th cells to these complexes activates the Th cells and causes them to divide.

Biological processes  
Apoptosis/decay: Each entity has a half-life, halfL. During each time step, this half-life is used to determine the probability, exp(-ln2/halfL), that a cellular entity undergoes apoptosis [2]. The probability of decay of non-cellular entities such as antibodies and cytokines is computed in a similar manner. Once an entity (cellular or non-cellular) undergoes apoptotic death, it is immediately removed from the simulation.

Resupply: To maintain a relatively constant population of all lymphocytes, these cells are continuously produced by the bone marrow and thymus to make up for those lost through apoptosis. During each time step, the half-life and original population size of each class of lymphocytes are used to determine the number of new cells of the given class to be added to the simulation.

Necrosis: During each time step, an infected cell undergoes necrosis or lysis if the number of intracellular viral particles it contains exceeds the viral burst size (i.e., the threshold of viral particles needed to elicit the lysis of an infected cell). Following necrosis, the cell releases new viral particles and is immediately removed from the simulation.

Cell Division: Only stimulated B, Th and Tc cells undergo cell division. A stimulated B cell divides over two time steps producing 4 new cells with the same receptors/specificity as the parent B cell. Fifty percent of the new cells become memory cells while the remaining 50% become plasma cells. The division of a stimulated Th (or Tc) cell follows similar rules with the difference being that 50% of the new cells become long-lived memory Th (or Tc) cells while the remaining 50% become effector Th (or Tc) cells. Note that dividing B cells may undergo hypermutation at a rate determined by the bMut parameter.

HIV infection: During a time step, each un-bound HIV particle is allowed to infect a randomly chosen neighboring cell. The probability of infection is determined by the infectivity of the HIV epitope. If infection occurs, the HIV disappears into the cytoplasm of the infected cell where it integrates its RNA into the cell's DNA. Infected cells present viral peptides on their surface, complexed with HLA1 molecules.

Viral replication: During each time step, HIV particles, found in infected cells, are allowed to replicate - each HIV divides once, producing two new viruses. Each new virus may mutate (i.e., it can be assigned a new randomly chosen (numerical) epitope and peptide). The probability of such mutation is determined by the vMut parameter.

CTL: This denotes the killing of infected cells by Tc cells. During each time step, a Tc cell is allowed to randomly kill one infected cell in its neighborhood/vicinity. The probability that a Tc cell will kill an infected cell on contact is determined by the parameter ctl-rate.

## HOW TO USE IT

The interface contains 10 sliders, two switches, six monitors, three plots, and the buttons "Setup" and "Run". 

Click the "Setup" button to initialize a new simulation and click "Run" to run the simulation.

The following information is reported during a simulation:  
The number of (1) non-viral entities (2) infected cells (3) viruses in the cytoplasms of productively infected cells, (5) neutralized HIV, and (6) infected cells killed by CTL. Also, the population size of the dominant HIV epitope and number of antibodies with high affinity (>70%) for this epitope is reported. This information can be used to determine what kind of therapeutic strategy (antibody or anti-retroviral) to simulate. In addition, the dynamics of antibodies, cytokines, B cells (naive, memory and plasma cells), Th cells (naive, memory and effector cells), Tc cells (naive, memory and effector cells), and HIV is plotted in real-time. 

While running a simulation, you can use the Command Center to perform the following:  
1) Simulate infection   
Simple type "infect specificity leth bsize repRate number," where "specificity" is the numerical value of each new virion's receptor and peptide, "leth" is the infectivity of each virion, "bsize" is the viral burst size, "repRate" is the viral replication rate, and "number" is the number of new virions you want to introduce into the simulation.

2) Simulate antibody therapy  
Simply type "Simulate-ABT specificity number" to introduce "number" new antibodies into the simulation. The value of "specificity" will be assigned to each antibody's paratope.

3) Simulate anti-retroviral therapy  
Simply type "Simulate-ART efficacy number" to introduce "number" new anti-retroviral agents (ara's) into the simulation. During each time step, the ara's thus introduced will be allowed to interact with a randomly chosen, neighboring virion and, if successful, inactivate the virion. The probability that an ara will inactivate a virion on contact is determined by the "efficacy" parameter.

The Sliders:  
1) numTh - the initial number of Th cells  
2) numB - the initial number of B cells  
3) numTc - the initial number of Tc cells  
4) numHIV - the initial number of HIV  
5) numApc - the initial number of APCs  
6) numEp - the initial number of epithelial cells  
7) infectivity - the represents the default infectivity of HIV - the probability that a virion will infect an cell on contact. Virions introduced after the simulation has begun can be assigned different infectivity.  
8) AICD - the rate of indirect killing of Th cells by HIV  
9) ctlRate - the probability that a Tc cell will kill an infected cell on contact. CTL rate is crucial to the control of HIV infection. However, it has been hypothesized that some Tc cells may be defective, suggesting that CTL rate is less that 1.0 (or 100%).  
10) neutRate - the rate at which an antibody neutralizes a virion, on contact.

Also, the values of about 30 global variables can be manipulated from the Procedures Tab.

The Switches:  
1) plots - use this to turn on/off real-time plots  
2) monitors - use this to turn/off updates to the monitors

The Plots:  
1) Antibodies - shows changes in the number of antibodies  
2) Cytokines - show changes in the number of cytokines  
3) Bcells - depicts the dynamics of B cells (includes naive, memory and effector or plasma cells)  
4) HIV/Th/Tc - depicts the dynamics of cell-free HIV, Th cells (includes naive, memory and effector cells), and Tc cells (includes naive, memory and effector cells). KEY: Red - HIV, Green - Tc, and Black - Th.

## THINGS TO TRY

Try to determine an optimal choice of parameter values for which each of the following occurs:  
1) Rapid (resp. slow) viral clearance.  
2) Rapid Th and Tc cell turnover.  
3) High/low antibody production.   
4) viral blips (periodic bursts of viraemia). (Hints: Decrease the value of pdissociation - follicular dendrytic cells, APCs in this case, may serve as viral reservoirs that help perpetuate chronic viraemia via occasional dissociation and the concomitant release of new virions [4]. Also, advanced users may want to implement anergic states for B, Th, and Tc cells)  
5) Depletion of Th cells (Hint: Manipulate the value of the AICD parameter)

Based on your observations, suggest putative critical parameters (e.g., viral infectivity) for the immune response to HIV.

## EXTENDING THE MODEL

HIVSIM is still far from being a comprehensive simulation of HIV immunodynamics. A number of other components of the immune response to HIV are yet to be implemented. Listed below are a few of them:  
1) B, Th, and Tc anergy  
2) Antibody-dependent cell-mediated cytotoxicity  
3) A better implementation of activation-induced cell death  
4) A better implementation of by-stander (cytokine-mediated) stimulation of B and Tc cells

In addition, further calibration with experimental data is necessary in order to make the simulation more robust and less sensitive to small changes in the values of some parameters.

## CREDITS AND REFERENCES

References  
1. McMichael A. J. and S. L. Rowland-Jones, 2001. "Cellular immune responses to HIV," Nature, 410:980-987.  
2. Kleinstein, S. H. and P. E. Seiden, 2000. "Simulating the immune system," Computing in Science and Engineering, pp.69-77.  
3. Silvestri, G. and M. B. Feinberg, 2003. "Turnover of lymphocytes and conceptual paradigms in HIV infection," J. Clin. Invest., 112:821-824.  
4. Hlavacek, W. S., N. I. Stilianakis, D. W. Notermansi, S. A. Danner, and A. S. Perelson, 2000. "Influence of follicular dendritic cells on decay of HIV during antiretroviral therapy," PNAS, 97(20):10966-10971.

Comments and suggestions should be sent to the author, Wilfred Ndifon, at wilfred@coursemate.net
@#$#@#$#@
default
true
0
Polygon -7500403 true true 150 5 40 250 150 205 260 250

ant
true
0
Polygon -7500403 true true 136 61 129 46 144 30 119 45 124 60 114 82 97 37 132 10 93 36 111 84 127 105 172 105 189 84 208 35 171 11 202 35 204 37 186 82 177 60 180 44 159 32 170 44 165 60
Polygon -7500403 true true 150 95 135 103 139 117 125 149 137 180 135 196 150 204 166 195 161 180 174 150 158 116 164 102
Polygon -7500403 true true 149 186 128 197 114 232 134 270 149 282 166 270 185 232 171 195 149 186
Polygon -7500403 true true 225 66 230 107 159 122 161 127 234 111 236 106
Polygon -7500403 true true 78 58 99 116 139 123 137 128 95 119
Polygon -7500403 true true 48 103 90 147 129 147 130 151 86 151
Polygon -7500403 true true 65 224 92 171 134 160 135 164 95 175
Polygon -7500403 true true 235 222 210 170 163 162 161 166 208 174
Polygon -7500403 true true 249 107 211 147 168 147 168 150 213 150

arrow
true
0
Polygon -7500403 true true 150 0 0 150 105 150 105 293 195 293 195 150 300 150

bee
true
0
Polygon -1184463 true false 152 149 77 163 67 195 67 211 74 234 85 252 100 264 116 276 134 286 151 300 167 285 182 278 206 260 220 242 226 218 226 195 222 166
Polygon -16777216 true false 150 149 128 151 114 151 98 145 80 122 80 103 81 83 95 67 117 58 141 54 151 53 177 55 195 66 207 82 211 94 211 116 204 139 189 149 171 152
Polygon -7500403 true true 151 54 119 59 96 60 81 50 78 39 87 25 103 18 115 23 121 13 150 1 180 14 189 23 197 17 210 19 222 30 222 44 212 57 192 58
Polygon -16777216 true false 70 185 74 171 223 172 224 186
Polygon -16777216 true false 67 211 71 226 224 226 225 211 67 211
Polygon -16777216 true false 91 257 106 269 195 269 211 255
Line -1 false 144 100 70 87
Line -1 false 70 87 45 87
Line -1 false 45 86 26 97
Line -1 false 26 96 22 115
Line -1 false 22 115 25 130
Line -1 false 26 131 37 141
Line -1 false 37 141 55 144
Line -1 false 55 143 143 101
Line -1 false 141 100 227 138
Line -1 false 227 138 241 137
Line -1 false 241 137 249 129
Line -1 false 249 129 254 110
Line -1 false 253 108 248 97
Line -1 false 249 95 235 82
Line -1 false 235 82 144 100

bird1
false
0
Polygon -7500403 true true 2 6 2 39 270 298 297 298 299 271 187 160 279 75 276 22 100 67 31 0

bird2
false
0
Polygon -7500403 true true 2 4 33 4 298 270 298 298 272 298 155 184 117 289 61 295 61 105 0 43

boat1
false
0
Polygon -1 true false 63 162 90 207 223 207 290 162
Rectangle -6459832 true false 150 32 157 162
Polygon -13345367 true false 150 34 131 49 145 47 147 48 149 49
Polygon -7500403 true true 158 33 230 157 182 150 169 151 157 156
Polygon -7500403 true true 149 55 88 143 103 139 111 136 117 139 126 145 130 147 139 147 146 146 149 55

boat2
false
0
Polygon -1 true false 63 162 90 207 223 207 290 162
Rectangle -6459832 true false 150 32 157 162
Polygon -13345367 true false 150 34 131 49 145 47 147 48 149 49
Polygon -7500403 true true 157 54 175 79 174 96 185 102 178 112 194 124 196 131 190 139 192 146 211 151 216 154 157 154
Polygon -7500403 true true 150 74 146 91 139 99 143 114 141 123 137 126 131 129 132 139 142 136 126 142 119 147 148 147

boat3
false
0
Polygon -1 true false 63 162 90 207 223 207 290 162
Rectangle -6459832 true false 150 32 157 162
Polygon -13345367 true false 150 34 131 49 145 47 147 48 149 49
Polygon -7500403 true true 158 37 172 45 188 59 202 79 217 109 220 130 218 147 204 156 158 156 161 142 170 123 170 102 169 88 165 62
Polygon -7500403 true true 149 66 142 78 139 96 141 111 146 139 148 147 110 147 113 131 118 106 126 71

box
true
0
Polygon -7500403 true true 45 255 255 255 255 45 45 45

butterfly1
true
0
Polygon -16777216 true false 151 76 138 91 138 284 150 296 162 286 162 91
Polygon -7500403 true true 164 106 184 79 205 61 236 48 259 53 279 86 287 119 289 158 278 177 256 182 164 181
Polygon -7500403 true true 136 110 119 82 110 71 85 61 59 48 36 56 17 88 6 115 2 147 15 178 134 178
Polygon -7500403 true true 46 181 28 227 50 255 77 273 112 283 135 274 135 180
Polygon -7500403 true true 165 185 254 184 272 224 255 251 236 267 191 283 164 276
Line -7500403 true 167 47 159 82
Line -7500403 true 136 47 145 81
Circle -7500403 true true 165 45 8
Circle -7500403 true true 134 45 6
Circle -7500403 true true 133 44 7
Circle -7500403 true true 133 43 8

cell
true
0
Circle -7500403 true true 114 114 73

circle
false
0
Circle -7500403 true true 35 35 230

person
false
0
Circle -7500403 true true 155 20 63
Rectangle -7500403 true true 158 79 217 164
Polygon -7500403 true true 158 81 110 129 131 143 158 109 165 110
Polygon -7500403 true true 216 83 267 123 248 143 215 107
Polygon -7500403 true true 167 163 145 234 183 234 183 163
Polygon -7500403 true true 195 163 195 233 227 233 206 159

sheep
false
15
Rectangle -1 true true 90 75 270 225
Circle -1 true true 15 75 150
Rectangle -16777216 true false 81 225 134 286
Rectangle -16777216 true false 180 225 238 285
Circle -16777216 true false 1 88 92

spacecraft
true
0
Polygon -7500403 true true 150 0 180 135 255 255 225 240 150 180 75 240 45 255 120 135

thin-arrow
true
0
Polygon -7500403 true true 150 0 0 150 120 150 120 293 180 293 180 150 300 150

truck-down
false
0
Polygon -7500403 true true 225 30 225 270 120 270 105 210 60 180 45 30 105 60 105 30
Polygon -8630108 true false 195 75 195 120 240 120 240 75
Polygon -8630108 true false 195 225 195 180 240 180 240 225

truck-left
false
0
Polygon -7500403 true true 120 135 225 135 225 210 75 210 75 165 105 165
Polygon -8630108 true false 90 210 105 225 120 210
Polygon -8630108 true false 180 210 195 225 210 210

truck-right
false
0
Polygon -7500403 true true 180 135 75 135 75 210 225 210 225 165 195 165
Polygon -8630108 true false 210 210 195 225 180 210
Polygon -8630108 true false 120 210 105 225 90 210

turtle
true
0
Polygon -7500403 true true 138 75 162 75 165 105 225 105 225 142 195 135 195 187 225 195 225 225 195 217 195 202 105 202 105 217 75 225 75 195 105 187 105 135 75 142 75 105 135 105

wolf
false
0
Rectangle -7500403 true true 15 105 105 165
Rectangle -7500403 true true 45 90 105 105
Polygon -7500403 true true 60 90 83 44 104 90
Polygon -16777216 true false 67 90 82 59 97 89
Rectangle -1 true false 48 93 59 105
Rectangle -16777216 true false 51 96 55 101
Rectangle -16777216 true false 0 121 15 135
Rectangle -16777216 true false 15 136 60 151
Polygon -1 true false 15 136 23 149 31 136
Polygon -1 true false 30 151 37 136 43 151
Rectangle -7500403 true true 105 120 263 195
Rectangle -7500403 true true 108 195 259 201
Rectangle -7500403 true true 114 201 252 210
Rectangle -7500403 true true 120 210 243 214
Rectangle -7500403 true true 115 114 255 120
Rectangle -7500403 true true 128 108 248 114
Rectangle -7500403 true true 150 105 225 108
Rectangle -7500403 true true 132 214 155 270
Rectangle -7500403 true true 110 260 132 270
Rectangle -7500403 true true 210 214 232 270
Rectangle -7500403 true true 189 260 210 270
Line -7500403 true 263 127 281 155
Line -7500403 true 281 155 281 192

wolf-left
false
3
Polygon -6459832 true true 117 97 91 74 66 74 60 85 36 85 38 92 44 97 62 97 81 117 84 134 92 147 109 152 136 144 174 144 174 103 143 103 134 97
Polygon -6459832 true true 87 80 79 55 76 79
Polygon -6459832 true true 81 75 70 58 73 82
Polygon -6459832 true true 99 131 76 152 76 163 96 182 104 182 109 173 102 167 99 173 87 159 104 140
Polygon -6459832 true true 107 138 107 186 98 190 99 196 112 196 115 190
Polygon -6459832 true true 116 140 114 189 105 137
Rectangle -6459832 true true 109 150 114 192
Rectangle -6459832 true true 111 143 116 191
Polygon -6459832 true true 168 106 184 98 205 98 218 115 218 137 186 164 196 176 195 194 178 195 178 183 188 183 169 164 173 144
Polygon -6459832 true true 207 140 200 163 206 175 207 192 193 189 192 177 198 176 185 150
Polygon -6459832 true true 214 134 203 168 192 148
Polygon -6459832 true true 204 151 203 176 193 148
Polygon -6459832 true true 207 103 221 98 236 101 243 115 243 128 256 142 239 143 233 133 225 115 214 114

wolf-right
false
3
Polygon -6459832 true true 170 127 200 93 231 93 237 103 262 103 261 113 253 119 231 119 215 143 213 160 208 173 189 187 169 190 154 190 126 180 106 171 72 171 73 126 122 126 144 123 159 123
Polygon -6459832 true true 201 99 214 69 215 99
Polygon -6459832 true true 207 98 223 71 220 101
Polygon -6459832 true true 184 172 189 234 203 238 203 246 187 247 180 239 171 180
Polygon -6459832 true true 197 174 204 220 218 224 219 234 201 232 195 225 179 179
Polygon -6459832 true true 78 167 95 187 95 208 79 220 92 234 98 235 100 249 81 246 76 241 61 212 65 195 52 170 45 150 44 128 55 121 69 121 81 135
Polygon -6459832 true true 48 143 58 141
Polygon -6459832 true true 46 136 68 137
Polygon -6459832 true true 45 129 35 142 37 159 53 192 47 210 62 238 80 237
Line -16777216 false 74 237 59 213
Line -16777216 false 59 213 59 212
Line -16777216 false 58 211 67 192
Polygon -6459832 true true 38 138 66 149
Polygon -6459832 true true 46 128 33 120 21 118 11 123 3 138 5 160 13 178 9 192 0 199 20 196 25 179 24 161 25 148 45 140
Polygon -6459832 true true 67 122 96 126 63 144

@#$#@#$#@
NetLogo 5.1.0
@#$#@#$#@
@#$#@#$#@
@#$#@#$#@
@#$#@#$#@
@#$#@#$#@
default
0.0
-0.2 0 0.0 1.0
0.0 1 1.0 0.0
0.2 0 0.0 1.0
link direction
true
0
Line -7500403 true 150 150 90 180
Line -7500403 true 150 150 210 180

@#$#@#$#@
0
@#$#@#$#@
